לימוד מערכים באמצעות תכנות משחקים פשוטים.
כל שנה בעת הוראת נושא המערך אני נתקל בקושי למצוא רעיונות שונים לתרגול. התרגילים הנפוצים, בדיקה אם מערך ממויין, עולה, יורד וכו', אם אלכסוני המערך עונים על תנאים שונים ומשונים וכדומה, אינם מעניינים ואין בהם כל גורם מעורר מוטיבציה. 

בשנה האחרונה בה לימדתי נושא זה מצאתי דרך חדשה שונה לגמרי לתרגול הנושא. הרעיון עלה במקרה, כאשר תוך כדי אחד השעורים הראשונים בנושא מערכים, הבחנתי בתלמידה אשר שקועה יתר על המידה במסך המחשב. כאשר ניגשתי לבדוק מה מושך את התעניינותה התברר לי כי היא משחקת במשחק הנקרא Code-Break - פיצוח קוד, משחק המוכר גם בשם "בול-פגיעה", בו אחד המשחקים מרכיב רצף של ארבעה צבעים ועל המשחק השני לנסות ולנחש רצף זה, השחקן המנחש מסדר רצף צבעים, ועל השחקן בעל הקוד לענות במספר ה"בולים" וה"פגיעות" שיש ברצף זה. בול משמעו צבע נכון ובמקום הנכון, פגיעה משמעה צבע נכון אך לא במקום הנכון. כך לאחר סדרה של ניחושים ניתן לפצח את הקוד.

הצעתי לכיתה שנכתוב בעצמנו משחק כזה, והכיתה הגיבה בהתלהבות לרעיון, התלהבות שנמשכה גם במהלך הביצוע. מימוש המשחק נעשה תוך כדי שימוש במערכים והצריך קלט למערך השמה במערך וסריקת המערך.  השלב הבא היה כתיבה של המשחק "איקס עיגול" (כאשר המשחק נועד לשני משתתפים, ולא נגד המחשב) הלוח יוצג על ידי מערך דו מימדי של תווים שאותחל על ידי התו רווח, השתמשנו בתצוגת טקסט כאשר האותיות X ו O משמשות לייצוג המשחק, במשחק זה כבר היה צורך לבדוק גם אלכסונים. בשלב זה התלמידים בקשו להפוך את הפלט לגרפי. לשם כך לימדתי את הפונקציות הבסיסיות של גרפיקה, לצורך תרגול עשינו הפסקה מהמשחקים וכתבנו תוכנית שהציגה שעון אנלוגי על המסך. (תרגיל זה הצריך גם שימוש בפונקציות Sin  ו Cos) לאחר שהכירו את הפונקציות לציור קווים ועיגולים כל מה שנשאר היה לשנות את פרוצדורת הפלט, והמשחק איקס עיגול הפך לגרפי. התלמידים שהיו תלמידי פנימיה היו גאים מאוד להראות לחבריהם לחדר את המשחקים שכתבו (גם השעון הפך מקור ליצירתיות והם יצרו שעונים בצבעים שונים בצורות שונות ובפונטים שונים, התרגיל גרם להם ללמוד בעצמם פונקציות גרפיקה נוספות).

המשחק הבא שכתבנו היה "Snake" נחש הנע על פני לוח אוכל "תפוחים" וצריך להיזהר מלפגוע בקירות ובעצמו. גם משחק זה התחלנו בייצוג טקסטואלי ולאחר סיומו תוך שינוי פרוצדורת הפלט בלבד הפכנוהו לגרפי. כתרגיל בית ביקשתי מהתלמידים לכתוב משחק דמקה לשני משתתפים, הייתה זו הפעם הראשונה בה תלמידים פנו אלי עם שאלות ואף בקשו להראות לי את עבודתם, במהלך ההפסקות עוד לפני השיעור הבא.

המשחקים  הבאים היו "פזל" לוח מספרים מעורבבים שיש לסדרם לפי סדר ובו היה צורך לבדוק אם המערך הדו ממדי ממוין, ומשחק "טטריס" הידוע, לסיום כתבנו את המשחק "שולה המוקשים" במשחק זה גילוי השטחים שאינם ממוקשים הינו תהליך רקורסיבי, מכיוון שרקורסיה עדיין לא נלמדה בשלב זה, כתבתי את תהליך זה בעצמי והסברתיו לתלמידים גם כאן הרעיון נתפש ואני חושב כי עצם לימודו תוך כדי כתיבת תוכנית מאתגרת ומעניינת תרם להבנתו מאוחר יותר.

אלגורתמים למשחקים.

בול פגיעה

הקוד ייבנה מארבע ספרות בין אחד לשש (ניתן לצור קוד מורכב יותר אך לכל היותר בין 0 ל 9), כל ספרה תופיע פעם אחת בלבד. נשתמש בשני מערכים בני ארבעה תאים עבור הקוד שיבחר המחשב ועבור הניחוש של המשחק. בנוסף לצורך הכנת הקוד נשתמש במערך דגלים בן שש תאים לצורך סימון מספרים שנבחרו על מנת שלא נבחר אותם שוב (מערך זה נועד לצורך שיפור סיבוכיות הכנת הקוד).

generate_code( code[4])  {שגרה המחזירה מערך ובו קוד בן 4 ספרות שונות בין 1 ל 6}
1. for I( 1 to 6 do

1.1. flags[I]( 0;

2. for I( 1 to 4 do

2.1.   repeat 

2.1.1     n(random(6)+1;

2.1.2   until flags[n]= 0;

2.2   flags[n](1;

2.3   code[I](n;

3. return code;

count_bool (code [4], guess[4]);

1. bool (0; 

2. for I( 1 to 4 do

2.1.   if (code[I]=guess[I]) 

2.1.1        bool( bool+1;

3. return bool; 

count_pgia (code [4], guess[4]);

1. pgia (0; 

2. for I( 1 to 4 do

2.1.  for J( 1 to 4 do 

2.2.     if (code[I]=guess[J]) and (I<>J) 

2.2.1.        pgia( pgia+1;

3. return pgia; 

Game:
1. Generate_code(code);

2. repeat

2.1. for I( 1 to 4 do

2.1.1.    read(guess[I]);

2.2. bool( count_bool(code,guess);

2.3. pgia( count_pgia(code,guess);

2.4. write (bool,pgia);

3. until bool=4

4. end.

program bool_pgia;

uses crt;

type gamearr=array[1..4]of integer;

var

  guess,code:gamearr;

  i,bool,pgia:integer;

procedure generatecode(var c:gamearr);

var

 i,j:integer;

 flags:array [1..6] of boolean;

begin

   for i:=1 to 6 do flags [i]:=true;  (* true means valid number (not selected yet) *)

   c[1]:= random(6)+1;

   flags[c[1]]:=False;   (* this number will no longer be valid *)

   for i:=2 to 4 do

   begin

     repeat

       c[i]:=random(6)+1;

     until flags[c[i]];

     flags[c[i]]:=False;

   end;

end;

procedure getguess(var c:gamearr);

var i,j:integer;

begin

   write ('Enter your guess : ');

   readln (c[1],c[2],c[3],c[4]);

end;

function count_Bool(c,g:gamearr):integer;

var

 bool,i:integer;

begin

  Bool:=0;

  for i:=1 to 4 do

     if c[i]=g[i] then Bool:=Bool+1;

  count_Bool:=Bool;

end;

function count_Pgia(c,g:gamearr):integer;

var

 Pgia,i,j:integer;

begin

  Pgia:=0;

  for i:=1 to 4 do

   for j:=1 to 4 do

     if (c[i]=g[j])and (i<>j) then Pgia:=Pgia+1; 

(* if  i=j then it is a bool , not Pgia *)

  count_Pgia:=Pgia;

end;

begin

   clrscr;

   Randomize;

   generatecode(code);

   repeat

      writeln('+ = Bool ',#251,' = Pegia');

      getguess(guess);

      bool:=count_Bool(code,guess);

      pgia:=count_Pgia(code,guess);

      for i:=1 to bool do write ('+');

      for i:=1 to pgia do write (#251);

      writeln;

   until bool=4;

   readln;

end.

program tictactoe;

(************************************************************************

 * The X and O game, the purpose is to create a line of the same letter *

 * the game will end if there is a winner or if the bord is full        *

 ************************************************************************)

uses crt;

type gamebord=array[0..4,0..4] of char;

var

  game: gamebord;

  row,col,plaingchar:char; {for using getchar}

  i,j,count:integer;

  winning:boolean;

Procedure drawgame(game:gamebord);

var

  i,j:integer;

begin

    clrscr;

    writeln('+-+-+-+');

    for j:=1 to 3 do {draw the game}

    begin

      write('|');

      for i:=1 to 3 do write(game[i,j],'|');

      writeln;

      writeln('+-+-+-+');

    end;

end;

Begin

  winning:=false;

  count:=0;

  for i:=0 to 4 do

    for j:=0 to 4 do game[i,j]:=' ';

  plaingchar:='X';

  repeat

    drawgame(game);

    writeln ('Enter column 1-3 and raw 1-3');

    repeat

      col:=readkey; i:=ord(col)-ord('0');

      row:=readkey; j:=ord(row)-ord('0');

      if game[i,j]<>' ' then write(#7);

    until (game[i,j]=' ') and (i in [1..3]) and (j in [1..3]);

    count:=count+1;

    game[i,j]:=plaingchar;

for i:=1 to 3 do

       for j:=1 to 3 do

        if (game[i-1,j+1]=plaingchar) and (game[i,j]=plaingchar) and (game[i+1,j-1]=plaingchar) or

           (game[i-1,j-1]=plaingchar) and (game[i,j]=plaingchar) and (game[i+1,j+1]=plaingchar) or

           (game[i,j-1]=plaingchar) and  (game[i,j]=plaingchar) and(game[i,j+1]=plaingchar) or

           (game[i-1,j]=plaingchar) and  (game[i,j]=plaingchar) and(game[i+1,j]=plaingchar) then

          winning:=true;

    if plaingchar='X' then plaingchar:='O' else plaingchar:='X';

  until winning or (count = 9);

  drawgame(game);

  if winning then

  begin

    if plaingchar='X' then plaingchar:='O' else plaingchar:='X';

    writeln(plaingchar, ' Won the Game');

  end else

  writeln ('No Winner');

end.

program snake;

uses crt;

const

      size=20;

      Left='K'; Right='M'; Up='H'; Down='P'; (*Ascii Value of Errow keys *)

type

  gamebord=array[0..size+1,0..size+1] of char;

var

  game:gamebord;

  n,i,snake_i,apple_i,j,snake_j,apple_j,snakel,spead,applecount:integer;

  Dir:char;

  Fail:boolean;

procedure initfild(var fild:gamebord);

var

  i,j:integer;

begin

 for i:=1 to size do  { clear game surface }

   for j:=1 to size do

      fild[i,j]:=#0;

 for i:=0 to size+1 do { mark filed border }

 begin

   fild[0,i]:=#219;

   fild[size+1,i]:=#219;

   fild[i,0]:=#219;

   fild[i,size+1]:=#219;

 end;

end;  {of initfild }

procedure drawfild(fild:gamebord);

var

  i,j:integer;

begin

  clrscr;

  for  i:=0 to size+1 do

  begin

    for  j:=0 to size+1 do

     if ((game[i,j])=char(snakel))then write(#1)

      else

      if (game[i,j] in[#1..#50]) then write('*')

       else write(game[i,j]);

     writeln;

  end;

end;

begin

 randomize;

 Fail:=False;

 TextMode($3);

 initfild(game);

 applecount:=0;

 spead:=MAXINT; (* Initial delay, Decreases after every apple eatten *)

 Dir:=Left;     (* Initial snake direction *)

 snakel:=1;     (* Initial snake length *)

 snake_i:=size div 2; snake_j:=size div 2; (* Initial snake position *)

 game[snake_i,snake_j]:=char(snakel);

 game[random(size-2)+1,random(size-2)+1]:='@';

 repeat

      if keypressed then dir:=readkey;

      if (dir=#0) then dir:=readkey;

      case dir of            

        Left : dec(snake_j);

        Right: inc(snake_j);

        Up   : dec(snake_i);

        Down : inc(snake_i);

      end;

      case game[snake_i,snake_j] of

        '@'  : begin

               inc(snakel); spead:=spead-trunc(spead/20);


       repeat



    apple_i:=random(size-2)+1;



    apple_j:=random(size-2)+1;


       until (game[apple_i,apple_j]=#0);


       game[apple_i,apple_j]:='@';

               inc(applecount);

              end;

        #0  : for i:=1 to size do

              begin



    for j:=1 to size do



        if ((game[i,j]<>'@')and(game[i,j]>#0)) then




     dec(game[i,j]);

              end;

       else   fail:=true;

      end;

      game[snake_i,snake_j]:=char(snakel);

  drawfild(game);

  writeln ('You ate ',applecount,' Apples');

  delay(spead);

  until fail;

  writeln ('SORRY You Lost The Game');

readln;

end.

PROGRAM Clock;

uses Dos, crt, graph;

const

  HStrelka=100;

  MStrelka=130;

  SStrelka=150;

  hour=Pi*2/12;

  minuet=Pi*2/60;

var

  h, m, s,hund : Word;

  midx,midy,Gd, Gm,i: Integer;

  st,st1:string;

begin

 Gd := Detect;

 InitGraph(Gd, Gm, 'c:\tp\bgi');

 if GraphResult <> grOk then

   Halt(1);

 midx:=getmaxx div 2;

 midy:=getmaxy div 2; 

 St:='';

 SetTextJustify(CenterText, CenterText);

   while (not keypressed) do

   begin

      GetTime(h,m,s,hund);

      setcolor(black);

      outtextxy(midx,midy+10,st);

      str(h,st1);

      St:='';

      St:=st+st1+':';

      str(m,st1);

      St:=st+st1+':';

      str(s,st1);

      St:=st+st1;

      setcolor(red);

      outtextxy(10,10,st);

      outtextxy(midx,midy+10,st);

      circle(midx,midy,midy);

      setcolor(red);

      line(midx,midy,

           trunc(midx - Hstrelka*cos(hour*(h-12)+0.5*pi)),

           trunc(midy - Hstrelka*sin(hour*(h-12)+0.5*pi)));

      setcolor(yellow);

      line(midx,midy,

           trunc(midx - Sstrelka*cos(minuet*(s)+0.5*pi)),

           trunc(midy - Sstrelka*sin(minuet*(s)+0.5*pi)));

      setcolor(green);

      line(midx,midy,

           trunc(midx - Mstrelka*cos(minuet*(m)+0.5*pi)),

           trunc(midy - Mstrelka*sin(minuet*(m)+0.5*pi)));

      delay(900);

      clearviewport;

   end;

end.

עותק ממסמך זה וכן התוכניות בפסקל ניתנים להורדה מהאינטרנט בכתובות הבאות:

המסמך:
http://www.geocities.com/yokylevy/pub/arrgames.doc

התוכניות בפסקל:
 http://www.geocities.com/yokylevy/pub/arrgames.zip

דוגמאות נוספות בשפת C:
http://www.geocities.com/yokylevy/pub/my_c_demos.zip

